NAG Library Routine Document

S15AEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S15AEF returns the value of the error function $\mathrm{erf}(x)$, via the function name.

2 Specification

```
FUNCTION S15AEF (X, IFAIL)
REAL (KIND=nag_wp) S15AEF
INTEGER            IFAIL
REAL (KIND=nag_wp) X
```

3 Description

S15AEF calculates an approximate value for the error function

$$\mathrm{erf}(x) = \frac{2}{\sqrt{\pi}}\int_0^x e^{-t^2}\,dt = 1 - \mathrm{erfc}(x).$$

Let $\hat{x}$ be the root of the equation $\mathrm{erfc}(x) - \mathrm{erf}(x) = 0$ (then $\hat{x} \approx 0.46875$). For $|x| \leq \hat{x}$ the value of $\mathrm{erf}(x)$ is based on the following rational Chebyshev expansion for $\mathrm{erf}(x)$:

$$\mathrm{erf}(x) \approx xR_{\ell,m}\left(x^2\right),$$

where $R_{\ell,m}$ denotes a rational function of degree ℓ in the numerator and m in the denominator.

For $|x| > \hat{x}$ the value of $\mathrm{erf}(x)$ is based on a rational Chebyshev expansion for $\mathrm{erfc}(x)$: for $\hat{x} < |x| \leq 4$ the value is based on the expansion

$$\mathrm{erfc}(x) \approx e^{x^2}R_{\ell,m}(x);$$

and for $|x| > 4$ it is based on the expansion

$$\mathrm{erfc}(x) \approx \frac{e^{x^2}}{x}\left(\frac{1}{\sqrt{\pi}} + \frac{1}{x^2}R_{\ell,m}\left(1/x^2\right)\right).$$

For each expansion, the specific values of ℓ and m are selected to be minimal such that the maximum relative error in the expansion is of the order 10^{-d}, where d is the maximum number of decimal digits that can be accurately represented for the particular implementation (see X02BEF).

For $|x| \geq x_{\mathrm{hi}}$ there is a danger of setting underflow in $\mathrm{erfc}(x)$ (the value of x_{hi} is given in the Users' Note for your implementation). For $x \geq x_{\mathrm{hi}}$, S15AEF returns $\mathrm{erf}(x) = 1$; for $x \leq -x_{\mathrm{hi}}$ it returns $\mathrm{erf}(x) = -1$.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Cody W J (1969) Rational Chebyshev approximations for the error function *Math.Comp.* **23** 631–637

5 Parameters

1: X – REAL (KIND=nag_wp) *Input*

On entry: the argument x of the function.

2: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

There are no failure exits from S15AEF. The parameter IFAIL has been included for consistency with other routines in this chapter.

7 Accuracy

See Section 7 in S15ADF.

8 Parallelism and Performance

Not applicable.

9 Further Comments

None.

10 Example

This example reads values of the argument x from a file, evaluates the function at each value of x and prints the results.

10.1 Program Text

```
    Program s15aefe

!     S15AEF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, s15aef
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: x, y
      Integer                          :: ifail, ioerr
!     .. Executable Statements ..
      Write (nout,*) 'S15AEF Example Program Results'

!     Skip heading in data file
```

```
      Read (nin,*)

      Write (nout,*)
      Write (nout,*) '     X           Y'
      Write (nout,*)

data: Do
        Read (nin,*,Iostat=ioerr) x

        If (ioerr<0) Then
          Exit data
        End If

        ifail = 0
        y = s15aef(x,ifail)

        Write (nout,99999) x, y
      End Do data

99999 Format (1X,1P,2E12.3)
    End Program s15aefe
```

10.2 Program Data

```
S15AEF Example Program Data
           -6.0
           -4.5
           -1.0
            1.0
            4.5
            6.0
```

10.3 Program Results

```
 S15AEF Example Program Results

      X           Y

   -6.000E+00  -1.000E+00
   -4.500E+00  -1.000E+00
   -1.000E+00  -8.427E-01
    1.000E+00   8.427E-01
    4.500E+00   1.000E+00
    6.000E+00   1.000E+00
```